

Основные понятия квантовой механики

Классическая механика и электродинамика при попытке применить их к объяснению атомных явлений приводят к результатам, находящимся в резком противоречии с опытом. Глубокое противоречие теории с экспериментом свидетельствует о том, что построение теории, применимой к атомным явлениям, требует фундаментального изменения в основных классических представлениях и законах. В данной главе мы рассмотрим несколько примеров использования новых базовых принципов квантовой механики, следуя первой главе книги Ланау и Лифшица.

• Волновой пакет в случае свободного движения

Напомним, что при переходе от квантовой к классической механике движение, описываемое волновой функцией, в общем случае отнюдь не переходит в движение по определенной траектории. Для того чтобы получить движение по определенной траектории, надо исходить из волновой функции особого вида, заметно отличной от нуля лишь на очень малом участке пространства (так называемый волновой пакет). В квазиклассическом пределе размеры этого участка надо уменьшать вместе с константой Планка. Тогда можно утверждать, что в квазиклассическом пределе волновой пакет будет перемещаться в пространстве по классической траектории частицы.

Следуя задаче N17 из задачника З.Флюгге, изучим волновой пакет в случае свободного движения частицы. Рассмотрим одномерный волновой пакет наиболее общего вида

$$\psi(x, t) = \int_{-\infty}^{\infty} \psi(k, x, t) dk \quad \psi(k, x, t) = C(k) e^{i(kx - \omega t)} \quad \omega = \frac{\hbar k^2}{2m}$$

где и

Для того, чтобы интеграл сходился, амплитуда $C(k)$ должна стремиться к нулю при $k \rightarrow \infty$, по крайней мере как $1/k$. Всякая выбранная подходящим образом амплитуда приводит к решению определенного вида.

Построим волновой пакет таким образом, чтобы в начальный момент $t = 0$ вероятность обнаружить описываемую им частицу заметно отличалась от нуля лишь внутри малой окрестности точки $x = 0$ и чтобы частица двигалась с импульсом $p_0 = \hbar q$ (не забывая про принцип неопределенности). Из этих условий следует, что в данном частном случае

$$\psi(x, 0) = A e^{\left(-\frac{x^2}{2a^2} + i q x \right)}$$

Разложение этой функции по плоским волнам имеет вид

$$\psi(x, 0) = \int C(k) e^{i k x} dk \quad C(k) = \frac{A a}{\sqrt{2\pi}} e^{\left(-\frac{(k-q)^2}{2} \right)}$$

, где .

Вычислив коэффициент $C(k)$ в начальный момент времени $t=0$ мы можем построить волновой пакет в любой момент времени

$$\psi(x, t) = \frac{A a}{\sqrt{2 \pi}} \int_{-\infty}^{\infty} e^{\left(-\frac{a^2 (k-q)^2}{2} + I k x - \frac{1 I \hbar k^2 t}{2 m} \right)} dk$$

Вычислим этот интеграл и посмотрим на его поведение во времени. Попробуем проделать все дальнейшие вычисления из задачника вместе с Maple. Зададим аргумент экспоненты под интегралом

> restart;

arg:=-a^2/2*(k-q)^2+I*k*x-I*h*k^2/(2*m)*t;

$$\text{arg} := -\frac{1}{2} a^2 (k-q)^2 + I k x + \frac{-\frac{1}{2} I \hbar k^2 t}{m}$$

Попробуем вычислить этот интеграл

> otv:=int(exp(arg),k=-infinity..infinity);

$$\text{otv} := \begin{cases} \frac{e^{\left(\frac{\frac{1}{2} I (-a^2 q^2 \hbar t + 2 m a^2 q x + I m x^2)}{a^2 m + I \hbar t} \right)} \sqrt{2} \sqrt{\pi}}{\sqrt{\frac{a^2 m + I \hbar t}{m}}} & \text{csgn}\left(\frac{1}{2} a^2 + \frac{\frac{1}{2} I \hbar t}{m}\right) = 1 \\ \infty & \text{otherwise} \end{cases}$$

В данном случае ответ означает следующее утверждение:

$$\text{csgn}\left(\frac{1}{2} a^2 + \frac{1 I \hbar t}{2 m}\right) = 1$$

если $\frac{1}{2} a^2 + \frac{1 I \hbar t}{2 m} > 0$, то интеграл равен экспоненте, во всех других случаях интеграл расходится, т.е. "равен" бесконечности. Из этого сложного выражения можно выделить необходимый нам ответ, определив тип полученного нами выражения

> whatype(otv);

function

Для данного типа выражения можно подсчитать число операндов в выражении и их последовательность

> nops(otv);

op(otv);

$$\operatorname{csgn}\left(\frac{1}{2}a^2 + \frac{1}{2}lht\right) = 1, \frac{e^{\frac{\left(\frac{1}{2}l(-a^2q^2ht + 2ma^2qx + lmx^2)\right)}{a^2m + lht}}}{\sqrt{\frac{a^2m + lht}{m}}}, \infty$$

Таким образом, необходимый нам ответ - это второй операнд в выражении типа **функция**

> **op(2,otv);**

$$\frac{e^{\frac{\left(\frac{1}{2}l(-a^2q^2ht + 2ma^2qx + lmx^2)\right)}{a^2m + lht}}}{\sqrt{\frac{a^2m + lht}{m}}}$$

Естественно, что тот же самый ответ можно получить, следуя классическому подходу З. Флюгге. Далее мы повторим все вычисления из задачника для того, чтобы приобрести практический опыт работы с системами символьных вычислений. Начнем с упрощения аргумента экспоненты стоящей под интегралом

> **factor(arg);**

$$-\frac{\frac{1}{2}a^2mk^2 - 2a^2mkq + a^2mq^2 - 2lkxm + lhk^2t}{m}$$

Прямой подход не приводит к необходимому ответу и, поэтому, используем ряд встроенных процедур из библиотеки (пакета) **student**

> **arg:=student[completesquare](arg,k);**

$$\arg := -\frac{1}{2} \frac{(a^2m + lht) \left(k - \frac{m(a^2q + lx)}{a^2m + lht} \right)^2}{m} + \frac{\frac{1}{2}(-la^2q^2ht + 2lma^2qx - mx^2)}{a^2m + lht}$$

Аргумент разлагается на два слагаемых. Первое, квадратичное по k и второе слагаемое, которое не зависит от k .

Посмотрим, как Maple работает с подобными интегралами. Начнем с первой квадратичной по k части аргумента. Рассмотрим табличный интеграл подобного вида

> **In:=int(exp(-b^2*z^2),z=-infinity..infinity);**

$$In := \begin{cases} \frac{\operatorname{csgn}(b) \sqrt{\pi}}{b} & \operatorname{csgn}(b^2) = 1 \\ \infty & \text{otherwise} \end{cases}$$

Мы опять получили условие на знак квадрата параметра b . Однако, при дополнительных условиях $0 < b$

> **interface(showassumed=0);**

```
assume(b>0);
In:=unapply(int(exp(-b^2*z^2),z=-infinity..infinity),b);
```

$$In := b \rightarrow \frac{\sqrt{\pi}}{b}$$

этот интеграл легко вычислить

```
> In(c);
```

$$\frac{\sqrt{\pi}}{c}$$

Заметим, что в данном случае мы задали табличный интеграл в виде отображения, которое будет использоваться нами в дальнейшем.

Таким образом, для вычисления необходимого нам интеграла остается научиться заменять переменную интегрирования. Итак, выделим квадратичное слагаемое в аргументе

```
> arg1:=op(1,arg);
```

$$arg1 := -\frac{1}{2} \frac{(a^2 m + I h t) \left(k - \frac{m (a^2 q + I x)}{a^2 m + I h t} \right)^2}{m}$$

и определим новую переменную

```
> k0:=solve(arg1,k);
```

$$k0 := \frac{m (a^2 q + I x)}{a^2 m + I h t}, \frac{m (a^2 q + I x)}{a^2 m + I h t}$$

После этого произвести замену переменных можно следующей встроенной командой из пакета **student**

```
> In1:=student[changevar](k-k0[1]=y,Int(exp(arg1),k=-infinity..infinity),y);
```

$$In1 := \int_{-\infty}^{\infty} e^{\left(-\frac{1}{2} \frac{(a^2 m + I h t) y^2}{m} \right)} dy$$

Как видим, изменилось не только подынтегральное выражение, но и пределы интегрирования. Тем не менее, теперь можно вычислить этот интеграл, накладывая дополнительные ограничения на значения параметров.

Можно использовать и другой путь - просто выделить коэффициент

```
> assume(a>0,h>0,t>0,m>0);
b0:=normal(sqrt(arg1/(k-k0[1])^2));
```

$$b0 := \frac{1}{2} \sqrt{-\frac{2 a^2 m + 2 I h t}{m}}$$

и подставить его в табличный интеграл

> **In(b0);**

$$2 \frac{\sqrt{\pi}}{\sqrt{-\frac{2a^2m+2lht}{m}}}$$

Теперь рассмотрим вторую часть аргумента, которая не зависит от

> **arg2:=op(2,arg);**

$$\text{arg2} := \frac{1}{2} \frac{-la^2q^2ht+2lma^2qx-mx^2}{a^2m+lht}$$

Выделим явно зависимость от переменной

> **arg2:=student[completesquare](arg2,x);**

$$\text{arg2} := -\frac{1}{2} \frac{m(x-la^2q)^2}{a^2m+lht} - \frac{1}{2} a^2q^2$$

Итак, после всех подготовительных работ, окончательный результат имеет вид

> **psi:=A*a/sqrt(2*Pi)*In(b0)*exp(arg2);**
psi:=normal(psi);

$$\psi = \frac{Aa\sqrt{2}e^{\left(\frac{1}{2} \frac{-la^2q^2ht+2lma^2qx-mx^2}{a^2m+lht}\right)}}{\sqrt{-\frac{2a^2m+2lht}{m}}}$$

После вычисления волновой функции волнового пакета, изучим поведение плотности волнового пакета

$$\rho(x,t) := |\psi(x,t)|^2$$

, которая равна

> **rho:=evalc(simplify(abs(psi)^2));**

$$\rho := \frac{a^2me^{\left(-\frac{a^2m^2x^2}{a^4m^2+h^2t^2} + \frac{ht(-a^2q^2ht+2ma^2qx)}{a^4m^2+h^2t^2}\right)}A^2}{\sqrt{a^4m^2+h^2t^2}}$$

Заметим, что в данном случае мы вынуждены "заставить" систему Maple произвести комплексные вычисления, используя

evalc

встроенную команду

. Так как в выражении для плотности вероятности

ρ

аргумент экспоненты достаточно

сложен, то его необходимо упростить

> rho:=simplify(rho);

$$\rho := \frac{a^2 m e^{\left(-\frac{a^2 (m x - q h t)^2}{a^4 m^2 + h^2 t^2} \right)} A^2}{\sqrt{a^4 m^2 + h^2 t^2}}$$

Детальное обсуждение полученного результата может быть найдено в задачнике Флюгге.

Теперь мы рассмотрим плотность потока вероятности для волнового пакета, которое имеет вид

$$s(x, t) = \frac{\hbar}{2 m i} \left[\overline{(\psi)} \left(\frac{\partial}{\partial x} \psi \right) - \psi \left(\frac{\partial}{\partial x} \overline{(\psi)} \right) \right]$$

Перепишем это определение в системе Maple в абстрактном виде

> s:=phi->(h/(2*I*m))*(conjugate(phi)*diff(phi,x)-phi*diff(conjugate(phi),x));

$$s := \phi \rightarrow \frac{-\frac{1}{2} \hbar \left(\overline{(\phi)} \left(\frac{\partial}{\partial x} \phi \right) - \phi \left(\frac{\partial}{\partial x} \overline{(\phi)} \right) \right)}{m}$$

Подставляя в данное определение волновую функцию волнового пакета, мы можем вычислить плотность

Однако, если рассматриваемого нами волнового пакета мы вычислим производные волновых функций

> assume(q,real); assume(x,real);

> simplify(diff(psi,x)/psi);

$$\frac{m (1 a^2 q - x)}{a^2 m + \hbar t}$$

> simplify(diff(conjugate(psi),x)/conjugate(psi));

$$\frac{m (1 a^2 q + x)}{-a^2 m + \hbar t}$$

то можно видеть, что производные пропорциональны самим функциям. Это означает, что в данном случае поток

пропорционален плотности $s = v \rho$, а коэффициент v имеет вид

> v:=simplify(s(psi)/rho);

$$v := - \frac{h |A|^2 (a^4 q m + x h t)}{A^2 (a^2 m + l h t) (-a^2 m + l h t)}$$

Заметим так же, что без дополнительных предположений о переменных q и x обозримый ответ получить практически невозможно. Далее заставим систему упростить данное выражение:

```
> v:=simplify(evalc(v));
```

$$v := \frac{h (a^4 q m + x h t)}{a^4 m^2 + h^2 t^2}$$


Для того, чтобы посмотреть на дисперсию скорости на графике, подставим частные значения параметров задачи в данное выражение для величины v

```
> v1:=simplify(subs(h=1,m=1,a=1,q=2,v));
```

$$v1 := \frac{2 + x t}{1 + t^2}$$


Далее нарисуем зависимость отношения плотности потока вероятности к вероятности, т.е. зависимость "скорости" от координаты в различные моменты времени

```
> p0:=plot(subs(t=0,v1),x=-5..10,color=black,
title='Линейность дисперсии', legend='t=0'):
p1:=plot(subs(t=1,v1),x=-5..10,color=red,legend='t=1'):
p2:=plot(subs(t=2,v1),x=-5..10,color=blue, legend='t=2'):
p3:=plot(subs(t=3,v1),x=-5..10,color=green, legend='t=3'):
plots[display](p0,p1,p2,p3);
```



Данные прямые линии характеризуют линейную дисперсию скорости. Рассмотрим, как изменяется распределение скорости во времени (анимация)

```
> plots[animate](v1,x=-5..10,t=0..5,frames=30,color=blue,title='Изменение скоросто во времени');
```



Достаточно сложное изменение скорости связано с тем, что различные части волнового пакета двигаются с разной скоростью, при этом часть волнового пакета даже движется в обратном направлении, т.е. с отрицательной скоростью.

Рассмотрим на связанное с этим стандартное "расплывание" волнового пакета. Подставим частные значения параметров в выражение для плотности


$$\rho(x, t)$$

```
> rhoS:=simplify(subs(A=(Pi)^(-1/4),h=1,m=1,a=1,q=2,rho));
```

$$\rho S := \frac{e^{-\frac{(-x+2t)^2}{1+t^2}}}{\sqrt{1+t^2} \sqrt{\pi}}$$

Посмотрим, как изменяется со временем распределение плотности по координате (анимация)

```
> plots[animate](rhoS,x=-5..15,t=0..5,frames=30,color=red,numpoints=200,title='Изменение распределения плотности во времени');
```



Полная вероятность при этом не зависит от времени и равна единице


```
> int(rhoS,x=-infinity..infinity);
```

1

$$\log(\rho)$$

На наш взгляд более наглядно процесс "размазывания" волнового пакета виден на графике логарифма плотности (анимация)

```
> LogRho:=simplify(log(rhoS));
plots[animate](LogRho,x=-5..15,t=0..5,view=[-5..15,-10..0],frames=30,color=red,numpoints=200,title='Изменение логарифма плотности во времени');
```



Достаточно наглядно видно, что различные части волнового пакета двигаются с различными скоростями (даже по направлению), что и приводит к возрастанию неопределенности координаты частицы после процесса измерения в начальный

$$t := 0$$

момент времени .

- **Квазиклассическое приближение**

Проследим, каким образом происходит в уравнении Шредингера переход к классической механике, рассматривая для простоты всего одну частицу во внешнем поле. Подставим в уравнение Шредингера

> restart;

eq:=I*h*diff(psi(x,t),t)=-h^2/(2*m)*diff(psi(x,t),x\$2)+U(x,t)*psi(x,t);

$$eq := I \hbar \left(\frac{\partial}{\partial t} \psi(x, t) \right) = -\frac{1}{2} \frac{\hbar^2 \left(\frac{\partial^2}{\partial x^2} \psi(x, t) \right)}{m} + U(x, t) \psi(x, t)$$

следующее выражение для волновой функции

> phi:=A(x,t)*exp(I/h*S(x,t));

$$\phi := A(x, t) e^{\left(\frac{I S(x, t)}{\hbar} \right)}$$

получим

> eq1:=expand(subs({psi(x,t)=phi}, eq/phi));

$$eq1 := \frac{I \hbar \left(\frac{\partial}{\partial t} A(x, t) \right)}{A(x, t)} - \left(\frac{\partial}{\partial t} S(x, t) \right) = -\frac{1}{2} \frac{\hbar^2 \left(\frac{\partial^2}{\partial x^2} A(x, t) \right)}{A(x, t) m} + \frac{-I \hbar \left(\frac{\partial}{\partial x} A(x, t) \right) \left(\frac{\partial}{\partial x} S(x, t) \right)}{A(x, t) m} + \frac{-\frac{1}{2} I \hbar \left(\frac{\partial^2}{\partial x^2} S(x, t) \right)}{m} + \frac{\frac{1}{2} \left(\frac{\partial}{\partial x} S(x, t) \right)^2}{m} + U(x, t)$$

В этом уравнении имеются чисто действительные и чисто мнимые члены. Мы подразумеваем, что S и A - действительны, надо не забыть только сказать об этом и системе

> assume(A(x,t),real); assume(S(x,t),real); assume(m>0,h>0);

Приравнивая обе части в отдельности к нулю, получим два уравнения

> eq_r:=evalc(Re(rhs(eq1)))-evalc(Re(lhs(eq1)))=0;

$$eq_r := -\frac{1}{2} \frac{\hbar^2 \left(\frac{\partial^2}{\partial x^2} A(x, t) \right)}{A(x, t) m} + \frac{1}{2} \frac{\left(\frac{\partial}{\partial x} S(x, t) \right)^2}{m} + U(x, t) + \left(\frac{\partial}{\partial t} S(x, t) \right) = 0$$

> eq_i:=evalc(Im(rhs(eq1)))-evalc(Im(lhs(eq1)))=0;
eq_i:=expand(eq_i/h);

$$eq_i := -\frac{\left(\frac{\partial}{\partial x} A(x, t) \right) \left(\frac{\partial}{\partial x} S(x, t) \right)}{A(x, t) m} - \frac{1}{2} \frac{\frac{\partial^2}{\partial x^2} S(x, t)}{m} - \frac{\frac{\partial}{\partial t} A(x, t)}{A(x, t)} = 0$$

Пренебрегая в первом уравнении членом, содержащим $\hbar$, получим

> eq_HJ:=coeff(lhs(eq_r),h,0)=0;

$$eq_HJ := \frac{1}{2} \frac{\left(\frac{\partial}{\partial x} S(x, t) \right)^2}{m} + U(x, t) + \left(\frac{\partial}{\partial t} S(x, t) \right) = 0$$

известное классическое уравнение Гамильтона-Якоби для действия S . Мы видим, что при $\hbar \rightarrow 0$ классическая механика справедлива с точностью до величин первого, а не нулевого порядка по $\hbar$ включительно.

$$\rho = A(x, t)^2$$

Введем плотность вероятности нахождения частицы в том или ином местопроизраства. Тогда второе из полученных уравнений можно переписать в виде

> eq_rho:=Diff(rho(x,t),t)+(1/m)*Diff(rho(x,t)*Diff(S(x,t),x),x)=0;

$$eq_rho := \left(\frac{\partial}{\partial t} \rho(x, t) \right) + \frac{\frac{\partial}{\partial x} \rho(x, t) \left(\frac{\partial}{\partial x} S(x, t) \right)}{m} = 0$$

Проверим это

> expand(value(subs({rho(x,t)=A(x,t)^2},eq_rho) + eq_i*2*A(x,t)^2));

$$0 = 0$$

$$\frac{1}{m} \frac{\partial}{\partial x} S(x, t)$$

Это уравнение имеет наглядный физический смысл, так как $\frac{1}{m} \frac{\partial}{\partial x} S(x, t)$ есть классическая скорость v частицы. Поэтому второе уравнение eq_rho есть не что иное, как уравнение непрерывности, показывающее, что плотность вероятности "перемещается" по законам классической механики с классической скоростью v в каждой точке.

В 20-ые годы, т.е. в самом начале развития квантовой механики наряду с де-бройлевской волновой интерпретацией квантовой механики предпринимались так же попытки поместить квантовую механику в классическое прокрустово ложе. При этом первое полученное нами уравнение предлагалось рассматривать как классическое уравнение Гамильтона-Якоби с добавочным квантовым потенциалом

```
> Uq:=psi->-diff(psi,x$2)/2;
```

$$U_q := \psi \rightarrow -\frac{1}{2} \left(\frac{\partial^2}{\partial x^2} \psi \right)$$

отвечающем некоторой квантовой силе

```
> Fq:=psi->-diff(Uq(psi),x);
```

$$F_q := \psi \rightarrow -\left(\frac{\partial}{\partial x} U_q(\psi) \right)$$

Далее мы посмотрим как выглядит квантовый потенциал и квантовая сила в случае гауссовского волнового пакета, рассмотренного нами ранее

```
> f_g:=(x,t,alpha)->(1/sqrt(alpha*(1+I*t))*sqrt(Pi))*exp(-x^2/alpha^2/(1+I*t));
```


$$f_g := (x, t, \alpha) \rightarrow \frac{\sqrt{\pi} e^{\left(-\frac{x^2}{\alpha^2 (1+I t)} \right)}}{\sqrt{\alpha (1+I t)}}$$

В начальный момент времени данный пакет расположен в окрестности $x=0$ со средним моментом $p=0$. Посмотрим на изменение данного пакета со временем

```
> with(plots):
```


```
p1:=animate(Re(f_g(x,t,1)),x=-5..5,t=1..10,frames=10,numpoints=50,color=blue);p2:=animate(Im(f_g(x,t,1)),x=-5..5,t=1..10,frames=10,numpoints=50,color=red);
display([p1,p2]);
```

Warning, the name changecoords has been redefined



Здесь синим и красным цветом изображены вещественная и мнимая части данной функции. Посмотрим теперь как изменяется со временем плотность вероятности

```
> animate(Re(f_g(x,t,1))^2+Im(f_g(x,t,1))^2,x=-5..5,t=1..10,frames=10,numpoints=50,color=magenta);
```



Из предыдущих параграфов данной главы нам известно, что гауссовский пакет "расплывается" в координатном представлении, что мы и видели на графиках. При этом в импульсном представлении распределение не изменяется со временем. Напомним, что квантовый потенциал является мерой кинетической энергии, которая изменяется. Это изменение при постоянстве распределения импульсов должно быть связано с изменением фазы волновой функции $S(x,t)$.

Определим модуль гауссовского волнового пакета

> **A_g:=(x,t,alpha)->simplify(sqrt(evalc(conjugate(f_g(x,t,alpha))*f_g(x,t,alpha))));**

$$A_g := (x, t, \alpha) \rightarrow \text{simplify}(\sqrt{\text{evalc}((f_g(x, t, \alpha)) f_g(x, t, \alpha))})$$


Посмотрим что такое квантовая сила

> **simplify(Fq(A_g(x,t,alpha)));**

$$\frac{\sqrt{\pi} x \sim e^{\left(-2 \frac{x \sim^2}{\alpha^2 (1+t \sim^2)}\right)} \text{csgn}(\alpha) (-2 x \sim^2 + 3 \alpha^2 + 3 \alpha^2 t \sim^2)}{(1+t \sim^2)^{\left(\frac{13}{4}\right)} \alpha^7 \sqrt{\frac{e^{\left(-2 \frac{x \sim^2}{\alpha^2 (1+t \sim^2)}\right)} \text{csgn}(\alpha)}{\alpha}}}$$

и как она меняется со временем

> **animate(Fq((A_g(x,t,1))),x=-5..5,t=0..5,frames=5,numpoints=100);**




Видно, что убывание со временем очень быстрое. Тем не менее, распределение импульсов не меняется и это можно понять изучив изменение фазы волнового пакета

> **S_g:=(x,t,alpha)->simplify(evalc(arctan(Im(f_g(x,t,alpha))/Re(f_g(x,t,alpha)))));**

$$S_g := (x, t, \alpha) \rightarrow \text{simplify} \left(\text{evalc} \left(\arctan \left(\frac{\Im(f_g(x, t, \alpha))}{\Re(f_g(x, t, \alpha))} \right) \right) \right)$$

> `animate(S_g(x,t,1),x=-10..10,t=0..40,frames=40,numpoints=100);`




Нулевая в начальный момент времени фаза $S(x,t)$ затем расплывается, быстро осциллируя при этом. Для того, чтобы понять изменение кинетической энергии, которая включает в себя и изменение модуля и изменение фазы мы введем плотность кинетической энергии

> `tau:=(x,t,alpha)->simplify(-evalc(conjugate(f_g(x,t,alpha))*diff(f_g(x,t,alpha),x$2)/2));`

$$\tau := (x, t, \alpha) \rightarrow \text{simplify} \left(-\text{evalc} \left(\frac{1}{2} (f_g(x, t, \alpha)) \left(\frac{\partial^2}{\partial x^2} f_g(x, t, \alpha) \right) \right) \right)$$

Посмотрим как изменяются со временем вещественная и мнимая части кинетической энергии

> `p1:=animate(Re(tau(x,t,1)),x=-5..5,t=1..5,frames=10,numpoints=50,color=blue);`
 > `p2:=animate(Im(tau(x,t,1)),x=-5..5, t=1..5,frames=10,numpoints=50,color=red);`
 > `display([p1,p2]);`



Докажем теперь, что распределение импульсов при этом все-таки не меняется. Для того, чтобы вычислить матричный элемент $\langle p^2 \rangle$, который пропорционален интегралу от плотности кинетической энергии

> `evalf(int(tau(x,1,1),x=-infinity..infinity));`

1.968701243

> evalf(int(tau(x,2,1),x=-infinity..infinity));

1.968701243

> evalf(int(tau(x,4,1),x=-infinity..infinity));

1.968701243

Аналогично проверим сохранение следующего матричного элемента $\langle p^4 \rangle$

> p4:=(t,alpha)->int(evalc(conjugate(f_g(x,t,alpha))*diff(f_g(x,t,alpha),x\$4)),x=-infinity..infinity);

$$p4 := (t, \alpha) \rightarrow \int_{-\infty}^{\infty} \text{evalc} \left(\overline{(f_g(x, t, \alpha))} \left(\frac{\partial^4}{\partial x^4} f_g(x, t, \alpha) \right) \right) dx$$

> evalf(p4(0,1));

11.81220746

> evalf(p4(2,1));

11.81220746

> evalf(p4(4,1));

11.81220746

Таким образом, мы видим, что локальные изменения рассмотренных нами величин в координатном пространстве несут в себе слишком мало информации. Например, быстрые изменения фазы и амплитуды в координатном пространстве ничего не могут сказать о распределении моментов частиц в импульсном пространстве. Таким образом рассмотрение квантовой механики с помощью добавления квантового потенциала в классическую механику практически невозможно. Именно поэтому данная интерпретация квантовой механики была отвергнута в свое время. В следующем параграфе мы рассмотрим более адекватную интерпретацию квантовой механики и перехода ее к квазиклассическому пределу.

- **Принцип неопределенности.**

В то время как в классической механике в каждый данный момент частица обладает определенными координатами и скоростью, в квантовой механике дело обстоит совершенно иным образом. Действительно, получая в любой момент времени информацию о значении координаты и скорости, например, электрона, означало бы наличие определенной траектории, каковой электрон не обладает. Таким образом, в квантовой механике координаты и скорость электрона являются величинами, которые не могут быть одновременно точно измерены, т.е. не могут одновременно иметь определенных значений. Если

интервалы значений, в которых могут оказаться координата и импульс частицы обозначить Δx и Δp , то эти величины удовлетворяют **принципу неопределенности**

$$\hbar \leq \Delta x \Delta p$$

который был получен Гейзенбергом. Мы видим, что чем с большей точностью известна координата частицы (т.е. чем меньше Δx), тем больше неопределенность Δp . В частности, если частица находится в некоторой строго определенной точке пространства, то все значения импульса равновероятны. Наоборот, если частица имеет строго определенный импульс, то

равновероятны все ее положения в пространстве.

Далее мы воспользуемся представлением Де Бройля, в котором частица с импульсом p представляется плоской волной с частотой $\frac{E}{h}$ и длиной волны $\frac{h}{p}$. В этом случае частица с неопределенным импульсом представляется суперпозицией

волн (волновым пакетом). В данном примере мы рассмотрим гауссовский волновой пакет состоящий из N волн вида

$$\psi := \sum_{i=1}^N e^{\left(-\frac{(k_i - k_0)^2}{(2 \Delta k)^2} \right)} \sin \left(k_i x - \frac{h m k_i^2 t}{2} \right)$$

$$k_0 = \frac{p_0}{h}$$

Здесь $\frac{p_0}{h}$, а $\frac{p_0}{\delta k}$ наиболее вероятное значение импульса, а k_i волновые числа равномерно покрывающие некоторый интервал δk , такой что $\Delta k < \delta k$. Напомним, что согласно принципу неопределенности $1 \leq \Delta x \Delta k$.

Рассмотрим свойства данного волнового пакета при следующей средней скорости электрона $v_0 = 5 \cdot 10^5$ м/с.

> **restart; with(plots):**

Warning, the name changecoords has been redefined

Итак, определим значения константы Планка и массы электрона

> **hbar:=1.055*10⁻³⁴;**
m:= 9.1 * 10⁻³¹;

$$hbar := .1055000000 \cdot 10^{-33}$$

$$m := .9100000000 \cdot 10^{-30}$$

Зададим наиболее вероятное значение волнового числа

> **v:=5*10⁵;**
k[0]:= m*v/hbar;

$$v := 500000$$

$$k_0 := .4312796209 \cdot 10^{10}$$

и интервал неопределенности

> **delta_k := k[0]/50;**

$$\Delta k := .8625592418 \cdot 10^8$$

$$N = 40$$

Зададим число волн в пакете

```
> N:=40;
```

$$N := 40$$

$$k_i$$

$$\Delta k = 4 \Delta k$$

Предположим, что соответствующие волновые числа равномерно заполняют интервал


```
> kRange:=4*delta_k:
for i from 1 to N+1 do k[i]:=k[0]-kRange/2 + (i-1)*(kRange/N); od:
```

Построим теперь волновой пакет

```
> c1:=hbar/(2*m); c2:=1/(2*delta_k)^2:
psi:=0:
for i from 1 to N+1 do
psi:=psi+exp(-c2*(k[i]-k[0])^2)*sin(k[i]*x-c1*k[i]^2*t);
od:
```


Посмотрим, как выглядит данный волновой пакет в нулевой момент времени

```
> y:=subs(t=0,psi):
plot(y,x=0..4*10^(-8),title='Волновой пакет в начальный момент времени');
```



Используя анимацию, посмотрим как изменяется данный волновой пакет со временем, т.е. изучим движение волнового пакета

```
> animate(psi,x=0..3*10^(-8),t=0..6*10^(-14),frames=20,
numpoints=200,title='Движение волнового пакета');
```



$$k_i$$

Построим теперь другой волновой пакет, в котором соответствующие волновые числа равномерно заполняют интервал


$2 \Delta k$ $4 \Delta k$

ВМЕСТО

```
> kRange2:=2*delta_k:
for i from 1 to N+1 do
k[i]:=k[0]-kRange2/2+(i-1)*(kRange2/N); od:
psi2:=0:
for i from 1 to N+1 do
psi2:=psi2+exp(-c2*(k[i]-k[0])^2)*sin(k[i]*x-c1*k[i]^2*t);
od:
```

В начальный момент времени

```
> y:=subs(t=0,psi2):
plot(y,x=0..4*10^(-8),title='Волновой пакет в начальный момент времени');
```



данный пакет занимает "больше" места в пространстве, т.е. затухает медленнее с ростом значения координаты x . Таким образом мы видим явное следствие принципа неопределенности - чем уже интервал неопределенности для импульсов, тем шире интервал неопределенности для координаты.

Используя анимацию, посмотрим на движение данного волнового пакета

```
> animate(psi2,x=0..3*10^(-8),t=0..6*10^(-14),frames=20,
numpoints=200,title='Движение волнового пакета');
```

